

AGRODEP Technical Note TN-03

October 2012

Documentation of the Export Restrictions And import Tariffs Overall impacts (ERATO) PE model

Carmen Estrades, Yuan Gao and David Laborde

Contents

1		Introduction3		
2		Partial Equilibrium Model of Oilseed Value Chain: theoretical framework		
		2.1.1		Overview
		2.1.2		Supply side
		2.1.3	5	Demand side9
3		Mod	eling	g in GAMS
	3.	1	Data	11
	3.	2	Solv	er and model equations12
		3.2.1		What's MCP and why we use it12
		3.2.2	2	Define the complementarity relations13
		3.2.3	5	Declare the options for the model resolution17
		3.2.4	Ļ	Simulations of trade policy in the form of export taxes and import tariffs
		3.2.5	j	Interpret GAMS output
4		Арре	endix	A, GAMS code of main file <i>erato_mcp.gms</i> 25
5		Appendix B, GAMS code of simulation file <i>sim_mcp.gms</i> 3		
6		Reference		

List of Figures

Figure 1. Modeled structure of the oilseed value chain	4
Figure 2. Variables and Parameters Associated with the Production of Oilseed Value Chain	8

List of Tables

Table 1. Definitions of Subscripts, Parameters, Variables in Model and in GAMS	6
Table 2. File Structure in GAMS/MCP Modeling System for Oilseed Value Chain	11
Table 3. Complementary Items	14
Table 4. Descriptions of Base Case and Simulation Scenarios	18

1 Introduction

Partial equilibrium (PE) analysis is commonly used to analyze the effects of policies, in particular trade policies, on price and welfare in the market that is directly affected. The partial equilibrium model only depicts the economy in one particular sector or market. In the case of a single commodity PE, this approach neglects the indirect effects through which changes in the demand or supply for one good may influence the market for another good. Different from many single-commodity partial equilibrium models, the *Export Restrictions And import Tariffs Overall impacts (ERATO)* ¹model illustrates how a multimarket model can be used to analyze trade policy options in a context of a multi-product value chain. As an illustration, we examine the impacts of trade restrictions imposed along three stages of the oilseed value chain. The model can be easily adapted to other value chains, eg. Cotton, or expanded to include other oilseeds (eg. peanuts).

This application considers 11 regions: ten countries (Argentina, Brazil, Canada, China, European Union, India, Indonesia, Malaysia, Ukraine, and United States) and a restof-the-world aggregate are considered in the model, as they are the major producing, exporting, and importing countries of oilseed products; and four oilseeds. Nevertheless, the sectoral and regional coverage of the model can be modified easily in GAMS (the software of implementation) and the structure of input data is simplified using an Excel workbook.

This document details the theory and the practical choices done to develop this model.

2 Partial Equilibrium Model of Oilseed Value Chain: theoretical framework

2.1.1 Overview

This <u>global multimarket partial equilibrium</u> model is built based on the <u>three</u> stages of the oilseed production (Figure 1): (i) production of seeds, (ii) production of meals and oils, (iii) production of biodiesel.

As indicated in Figure 1, four types of seeds are included in the model, which are soybeans, sunflower seeds, rapeseeds, and palm nuts. The blue diagrams in Figure 1

¹ The ERATO model is a part of the MUlti Sectoral partial Equilibrium models (MUSE) develop for the AGRODEP consortium.

represent the supplies of different commodities, and the green diagrams in Figure 1 represent the final demands of different commodities. From the supply side, each type of seed, meals and oils are produced from the crushed seed, and biodiesel is converted from oil. Thus, the oil and meal supplies are a function of the crushed seed, and the biodiesel supply is a function of the oil. From the demand side, the demand of each commodity comes from both domestic market and international market. In the domestic market, we assume the livestock sector as a final demand of different types of meal as it is the main demand for meal. Oil demand comes from the final consumption by households and intermediate consumption by biodiesel sector. The final demand of biodiesel comes from the households. The total demand of each type of commodity is the sum of consumption and export.




Figure 1. Modeled structure of the oilseed value chain

This model assumes that each seed, meal, and oil is a homogeneous commodity in the international market. There is no product differentiation by country of origin (no Armington Assumption). There are three sets of equations for the partial equilibrium condition, which are the supply, demand, and international trade. The definitions of subscripts, parameters, and variables are summarized in Table 1. The important variables and parameters associated with the oilseed value chain production are indicated in Figure 2.

Subscripts/ Parameters/Variables/	Corresponding GAMS Item	Description
Subscripts		
i	i	type of seeds (soybeans, sunflower seeds, palm nuts, and rapeseeds)
S	type-seed	crushed seeds
0	type-oil	oils
m	type-meal	meals
b	type-bfuel	biodiesel
r	r, s	set of countries/regions
f	f	final demand (households and biodiesel producers)
Parameters		
$Ks_{i,r}$, $Ko_{f,r}$, $Km_{i,r}$, $Kb_{f,r}$	Ks(i,r), Ko(f,r), Km(i,f,r), Kb(f,r)	scale parameter in isoelastic supply/demand function
$A_{i,r}$	A(i,r)	scale parameter of crush demand
$\alpha_{o_i r} \alpha_{m_i r}$	Alpha(type,i,r)	technical coefficients of oil and meal production
$\varepsilon_{i,r}$	SIGA(i,r)	price elasticity of supply area
${ m emg}_{i,r}$, ${ m emgb}_r$	CRMG(i,r)	crush margin elasticity, biodiesel margin elasticity
Area _{ir}	Area(i,r)	area of land for harvesting
$aC_{i,f,r}$	a_C(i,f,r)	share parameter of different types of oils to the total composite oils
D _{0ir}	SIGO(I,f,r)	elasticity of substitution between different oils
$\sigma_{l_r}, \sigma_{m_{i_r}}, \sigma_{b_{i_r}}$	SIGL(r), SIGM(i,r), SIGB(r)	price elasticity of demand for oil, meal, and biodiesel
texp _{sir}	TEXP(type, i,r)	Ad valorem export taxes
timp _{si r}	TIMO(type, i,r)	Ad valorem import taxes
Variables		
$Ys_{i,r}$, $Ym_{i,r}$, $Yo_{i,r}$, Yb_r	Ys(i,r), Ym(i,r), Yl(i,r), Yb(r)	domestic production
$Xs_{i,r}$, $Xm_{i,r}$, $Xo_{i,r}$, Xb_r	Xs(i,r), Xm(i,r), Xl(i,r), Xb(r)	net trade
Cs_{ir} , Cm_{ir} , Cb_r	CR(i,r), Cm(i,f,r), Cb(f,r)	domestic demand of crushed seed, meal, and biorefinery
$Co_{i.f.r}$	Clb(i,r), Clh(i,r),	domestic demand of oil by biodiesel sector or households
Ccomp _{f,r}	CPb(r), CPh(r)	demand of composite oil by biodiesel sector or households
$Ps_{i,r}, Pm_{i,r}, Po_{i,r}, Pb_r$	Ps(i,r), Pm(i,r), Pl(i,r), Pb(r)	domestic price of seed, meal, oil and biodiesel
		-
Pcomp _{f,r}	PCb(r), PCh(r)	domestic price of composite oil by biodiesel or household
Ps*, Pm*, Po*, Pb*	PCIFs(i), PCIFm(i), PCIFl(i), PCIFb	international price of seed, meal, oil and biodiesel
MRG _{i,r} , MRGb _r	MRG(i,r), MRGb(r)	crush margin, biodiesel margin

Table 1. Definitions of Subscripts, Parameters, Variables in Model and in GAMS

Supply

$$Ys_{i,r} = Ks_{i,r}Area_{i,r}Ps_{i,r}^{\varepsilon_{i,r}}$$
(1)

$$Ym_{i,r} = \alpha_{m_{i,r}} Cs_{i,r} \tag{2}$$

$$Yo_{i,r} = \alpha_{o_{i,r}} Cs_{i,r} \tag{3}$$

$$MRG_{i,r} = (Po_{i,r}\alpha_{o_{i,r}} + Pm_{i,r}\alpha_{m_{i,r}}) - Ps_{i,r}$$
(4)

$$Cs_{i,r} = A_{i,r} \left(MRG_{i,r} \right)^{\varepsilon mg_{i,r}}$$
(5)

$$Yb_r = Ccomp_{f,r}, \forall f = biodiesel \ sector$$

$$MRGb_r = Pb_r - Pcomp_{f,r}, \ \forall f = biodiesel \ sector$$
(6)
(7)

$$Ccomp_{f,r} = Kb_{f,r}MRGb_r^{\ \varepsilon mgb_r} \ \forall f = biodiesel \ sector$$
(8)

Demand

$$Cm_{i,r} = Km_{i,r}Pm_{i,r}^{\sigma_{m_{i,r}}}$$
(9)

$$Ccomp_{f,r} = Ko_{f,r}Pcomp_{f,r}{}^{\sigma_{l_r}} \quad \forall f = households$$
(10)

$$Co_{i,f,r} = aC_{i,f,r}Ccomp_{f,r}\frac{Pcomp_{f,r}\sigma_{o_{i,r}}}{Po_{i,r}}$$
(11)

$$Pcomp_{f,r}Ccomp_{f,r} = \sum_{i} Po_{i,r}Co_{i,f,r}$$
(12)

$$Cb_r = Kb_r P b_r^{\sigma_{b_{i,r}}}$$
⁽¹³⁾

Trade

$$Xs_{i,r} + Cs_{i,r} = Ys_{i,r} \tag{14}$$

$$Xm_{i,r} + Cm_{i,r} = Ym_{i,r} \tag{15}$$

$$Xo_{i,r} + \sum_{f} Co_{i,f,r} = Yo_{i,r}$$
(16)

$$Xb_r + Cb_r = Yb_r \tag{17}$$

$$Ps^* = Ps_{i,r} \left(1 + texp_{s_{i,r}}\right) / (1 + timp_{s_{i,r}})$$
(18)

$$Pm^* = Pm_{i,r} \left(1 + texp_{m_{i,r}}\right) / (1 + timp_{m_{i,r}})$$
(19)

$$Po^* = Po_{i,r} \left(1 + texp_{o_{i,r}}\right) / (1 + timp_{o_{i,r}})$$
(20)

$$Pb^* = Pb_r (1 + texp_{b_r})/(1 + timp_{b_r})$$
(21)

$$\sum_{r} X s_{i,r} = 0 \tag{22}$$

$$\sum_{r} X m_{i,r} = 0 \tag{23}$$

$$\sum_{r} X m_{i,r} = 0 \tag{23}$$

$$\sum_{r} X o_{r,r} = 0 \tag{24}$$

$$\sum_{r} X b_{i,r} = 0 \tag{24}$$

$$\sum_{r} X b_{r} = 0 \tag{25}$$




Figure 2. Variables and Parameters Associated with the Production of Oilseed Value Chain

2.1.2 Supply side

There are six generic equations in the Supply section. The elasticity of supply is assumed to be constant during the production process in this model. Equation (1) is the production function for seed *i* in region $r(Ys_{i,r})$, which is a function of the price of seed $(Ps_{i,r})$ and the harvest area $(Area_{i,r})$. $Ks_{i,r}$ is a scale parameter. $\varepsilon_{i,r}$ represents the supply area elasticity of seed production. Yields are assumed to be constant. Then, the seeds are crushed into meals and oils (Figure 1 and Figure 2). Based on the Leontief technology assumption, a fixed proportion in the technology of meal and oil production from the crushed seeds is assumed. Equation (2) is the production function for meal i in region r $(Ym_{i,r})$ and Equation (3) is the production function for oil i region r $(Yo_{i,r})$. $Cs_{i,r}$ is the demand of crushed seed i in region r by both the meal and oil sectors. $\alpha_{m_{i,r}}$ and $\alpha_{o_{i,r}}$ are the fixed technical coefficients of crushed seeds in the production of meals and oils, respectively. Equation (4) is the crush margin ($MRG_{i,r}$, an input margin in per unit) equation which measures the value added of producing meals and oils from the crushed seed. The margins capture the cost of capital, labor and energy in the crushing process. It is derived from the formula $Cs_{i,r}(Ps_{i,r} + MRG_{i,r}) = (Po_{i,r}Yo_{i,r} + Pm_{i,r}Ym_{i,r})$ by replacing Equation (2) and (3) in it. Equation (5) defines the demand of crushed seeds $(Cs_{i,r})$ for seed i in region r, which is actual an isoelastic supply function of the seed volume produced. $Cs_{i,r}$ is depending on the crush margin ($MRG_{i,r}$). $A_{i,r}$ is a scale parameter for crush demand. $\epsilon m g_{i,r}$ is crush margin elasticity, which is the price elasticity of crushing seeds supply

based on the potential revenue (margins) of the crushing activity. In the third production stage, a composite vegetable oil is produced by blending crop-specific oil to meet the demand of final consumers (households) and the intermediate demand of biodiesel producers. Strong, but imperfect, substitution is assumed relying on a Constant Elasticity of Supply (CES) technology. It is assumed that one unit of the composite oil demanded by the biodiesel sector is transformed into one unit of biodiesel production (Figure 1). Thus, the production of biodiesel (*Yb_r*) is equivalent to its demand from the biodiesel sector (*Ccomp_{f,r}*), as indicated in Equation (6). Equation (7) represents the margin of biodiesel production (*MRGb_r*), which is determined by the differences between the domestic biodiesel price (*Pb_r*) and the domestic composite oil price (*Pcomp_{f,r}*). Similar with the demand of crushed seeds in Equation (5), the demand of composite oil by the biodiesel sector (*Ccomp_{f,r}* $\forall f = biodiesel sector$) in Equation (8) is also an isoelastic supply variable which depends on the biodiesel margin elasticity (*MRGb_r*). *Kb_{f,r}* is a scale parameter, and *ɛmgb_r* is the margin elasticity of composite oil production for biodiesel refinery, which is also the price elasticity of composite oil supply.

2.1.3 Demand side

In the demand section: Equation (9) is the final demand function for meal i in region r ($Cm_{i,r}$) by the livestock sector. $Km_{i,r}$ is a scale parameter, $Pm_{i,r}$ is the price of meals, and $\sigma_{m_{i,r}}$ is the price elasticity of meal demand. There is perfect substitution between different types of meals. Different types of oils are consumed by either the households or biodiesel sector, and the demands of composite oil from those two agents are isoelastic. Therefore, the demand for composite oil by households ($Ccomp_{f,r} \forall f = households$) is depicted in Equation (10). $Ko_{f,r}$ is the scale parameters. σ_{l_r} is the price elasticity of oil demand by households. Equations (10)-(12) are derived from the first order condition of utility maximization based on CES assumption, which ensures different types of oils are imperfect substitutes for both households and biodiesel producers. $aC_{i,f,r}$ is the share parameter of different types of oils to the total composite oils. $Co_{i,f,r}$ is the elasticity of substitution among different types of oils. Equation (13) represents the final demand of biodiesel by

9

households in region r (Cb_r), which is isoelastic and depends negatively on the price of biodiesel in domestic market (Pb_r). $\sigma_{b_{ir}}$ is the price elasticity of the biodiesel final demand.

Import and export activities are not explicitly shown in Figure 1 and Figure 2. However, this model assumed the total demand of each type of commodity is the sum of its domestic consumption and net exports, which inducing Equations (14)-(17). $Xs_{i,r}$, $Xo_{i,r}$, $Xm_{i,r}$, and Xb_r are the net exports of crushed seeds, meals, oils, and biodiesels based different types of seeds. Equations (18)-(21) are the international prices parity equations for different types of seeds. The international prices (P^*) are equal to the domestic prices (P) in the condition of without market distortion, in the form of export taxes ($texp_{i,r}$) or import tariffs ($timp_{i,r}$). Equations (22)-(25) represent the sum of net trade flows for each type of goods in region r in the international market ($\sum_r X_{i,r}$). As we assume the partial equilibrium is established in the international market for oilseed value chain, the sum of net export for each commodity market is equal to zero.

3 Modeling in GAMS

The theoretical model described in the previous section is implemented using the <u>GAMS</u> software. This section describes the different components of the applied model.

Table 2 summarizes the contents of the different files in the GAMS files package for the oilseed value chain partial equilibrium model. File "*erato_mcp.gms*" is the main file which includes all items in the model framework. File "*variablesbase.gms*" declares the main parameters used for baseline case of the MCP model. The .gdx format file "*data.gdx*" stores all initial values of parameters and variables appeared in the MCP model for different scenarios, it is converted from an Excel file ("*data_erato.xls*"). Gdx format has the advantages of large sparse data storage and easy data loading in GAMS compared to excel format. File "*varinit_mcp.gms*" and "*sim_mcp.gms*" are used to run the simulations of different trade policies. File "*results_mcp.gms*" is used to calculate the percentage changes of variables in simulations, so as to compare the model solution of each simulation scenario with that of the base case. Gdx file "*res_mcp.gdx*" stores the output of model solutions for all seven scenarios.

File name	Description
erato_mcp.gms	all items (sets, parameters, and variables) that appear in the model equations.
variablesbase.gms	define the parameters used for the baseline case.
data.gdx	store the initial values of parameters and variables before the new equilibrium is established.
varinit_mcp.gms	assign the initial values to variables for the purposes of running different scenarios of the trade policy simulation
sim_mcp.gms	simulation file, including one baseline scenario and seven scenarios of trade policy simulation
results_mcpf.gms	compare the output of simulation scenario with the base case
res_mcp.gdx	Store the output of model solutions

Table 2. File Structure in GAMS/MCP Modeling System for Oilseed Value Chain

3.1 Data

The data of commodity domestic production, domestic consumption, and trade are obtained from FAO-food balance sheet (2007). The data of import tariff are generated by Laborde (2010), and the data of export taxes are obtained from various sources, including Trade Policy Reports, WTO, and OECD. The parameters of price elasticity and crush margin elasticity are obtained from FAPRI elasticity database (2011). Due to the missing data for some countries, assumed values are assigned to certain parameters (Appendix A). In specific, a value of 0.3 is assigned to the missing data of price elasticity of supply area (*SIGA*(*i*,*r*)); a value of -0.3 is assigned to the missing data of the price elasticity of meal demand (*SIGM*(*i*,*r*)). Besides, the price elasticity of oil demand (*SIGL*(*r*)) for Argentina is assumed to be -0.28 in this study, while that for other countries are assumed to be 2. The price elasticity of substitution between different oils (*SIGO*(*i*,*f*,*r*)) is assumed to be 2. The price elasticity of biodiesel demand (*SIGB*(*r*)) is assumed to be -0.25. Since no data resource of biodiesel margin elasticity is currently available, the values of crush margin elasticity of soybean (*CRMG*(*"soy"*,*r*)) are used instead.

SIGA(i,r) = 0.3; SIGA(i,r) = elast(i,r,'Area'); SIGL(r) = -0.38;

```
SIGL('ARG') = -0.28;
SIGO(i,"h",r) = 2;
SIGO(i,"bio",r) = 2;
SIGM(i,r) = -0.3;
SIGM(i,r) = elast(i,r,'Mealp');
SIGB(r) = -0.25;
```

All the data are gathered in the *data_erato.xls* file. The contents of this file can be modified to accommodate new products, sectors or regions. In such case, the "sets" section of the *erato_mcp.gms* file has to be modified accordingly.

3.2 Solver and model equations

3.2.1 What's MCP and why we use it

We set up a model based on the framework of partial equilibrium model of oilseed value chain, and implement it as a Mixed Complementarity Problem (MCP) in GAMS language. MCP is a generalization of Non-linear complementarity problem (NCP). The mixed form of a complementarity condition involves a bounded variable $L_i \leq Z_i \leq U_i$ with an associated function $F_i(Z)$, it can be mathematically defined as (McCarl 2011):

solve for Z such that

$$F_{i}(Z) = 0 \text{ and } L_{i} \leq Z_{i} \leq U_{i}, \text{ or}$$

$$F_{i}(Z) \geq 0 \text{ and } Z_{i} = L_{i} , \text{ or}$$

$$F_{i}(Z) \leq 0 \text{ and } Z_{i} = U_{i}$$

$$(26)$$

where Z_i is a set of variables to be set; F_i is a function; L_i and U_i are a set of upper and lower bounds on the variables (L_i may be –inf and U_i may be +inf). The number of Z_i variables and the number of relations $F_i(Z)$ is equal. The above condition (26) states that either Z_i must reach one of its bounds, or the function $F_i(Z)$ must be zero. This is critically important in developing efficient solution algorithms. Condition (26) can be written compactly as

$$F(Z) \perp L \le Z \le U \tag{27}$$

where the \perp ("perp") symbol indicates pair-wise complementarity between the function F(Z) and the variable Z and its bounds. The MCP problem does not have an objective function.

The global multimarket partial equilibrium model in this study could be formulated into classic non-linear programs (NLP). However, a MCP model has the advantages of incorporating the features found in agricultural mathematical programming model with

12

inequalities (for example, agriculture supply, foreign trade, and policy flexibility), while those features cannot readily be captured in strict-equality simultaneous equation systems (Lofgren and Robinson 1997). Also, computational evidence suggests that algorithms for solving MCP are relatively reliable and efficient, particularly for models which are not natural optimization problems (Rutherford 1995). Path is a large-scale MCP solver available through GAMS, and it is used to solve the MCP model in this study.

3.2.2 Define the complementarity relations

Cast as a multimarket partial equilibrium problem for oilseed value chain, a system of inequalities/equations and associated complementary slackness conditions is employed to drive the market net trade flow to zero. To solve the mixed complementarity problems, the model must satisfy the following conditions in the model statement:

- (1) A consistent set of non-negative variables and inequalities where each variable is of the same dimension as associated inequality that it is complementary with,
- (2) A consistent set of unbounded variables and equations where each equation is complementary with each unbounded variable,
- (3) A model where no variable is complimentary with more than one inequality or equation.

The following model declaration statement in GAMS file "*erato_mcp.gms*" expresses the above conditions by using period to associate each variable with its complementary inequality/equation:

```
Model OSDS
/
EQ_Ys.Ys
EQ_Ym.Ym
EQ_Yo.Y1
EQ_MG.MRG
EQ_CR.CR
EQ_Yb.Yb
EQ_MGb.MRGb
EQ_CPb.CPb
EQ_CPb.CPb
EQ_CPh.CPh
EQ_Cob.Clb
EQ_Coh.Clh
EQ_PCb.PCb
```

EQ_PCh.PCh EQ_Cb.Cb EQ_Ps.Ps EQ_Pm.Pm EQ_Po.Pl EQ_Pb.Pb EQ_MMs.PCIFs EQ_MMm.PCIFm EQ_MMo.PCIF1 EQ_MMb.PCIFb EQ_Xs.Xs EQ_Xm.Xm EQ_Xo.X1 EQ_Xb.Xb /;

All decision variables except net trade volume (e.g. Xs, Xl, Xm, and Xb) are positively defined, as they must be restricted to be positive to make economic sense (e.g. production, consumption, and price). Specifically, the complementary relations between variables and inequalities/equations are shown in the Table 3 below. Positive variables are paired with inequalities, so the dimensions of the items are the same.

Positive Variables In GAMS	Inequality Name in GAMS	Equation Number in Model Framework
Ys(i,r)	EQ_Ys(i,r)	(1)
Ym(i,r)	EQ_Ym(i,r)	(2)
Yl(i,r)	EQ_Yo(i,r)	(3)
MRG(i,r)	EQ_MG(i,r)	(4)
CR(i,r)	EQ_CR(i,r)	(5)
Yb(r)	EQ_Yb(r)	(6)
MRGb(r)	EQ_MGb(r)	(7)
CPb(r)	EQ_CPb(r)	(8)
Cm(i,f,r)	EQ_Cm(i,f,r)	(9)
CPh(r)	EQ_CPh(r)	(10)
Clb(i,r)	EQ_Cob(i,r)	(11)
Clh(i,r)	EQ_Coh(i,r)	(11)
PCh(r)	EQ_PCh(r)	(12)
PCb(r)	EQ_PCb(r)	(12)
Cb(f,r)	EQ_Cb(f,r)	(13)
Ps(i,r)	EQ_Ps(i,r)	(14)
Pm(i,r)	EQ_Pm(i,r)	(15)
Pl(i,r)	EQ_Po(i,r)	(16)
Pb(r)	EQ_Pb(r)	(17)
PCIFs(i)	EQ_Xs(i,r)	(18)
PCIFm(i)	EQ_Xm(i,r)	(19)
PCIFl(i)	EQ_Xo(i,r)	(20)
PCIFb	EQ_Xb(r)	(21)

Table 3. Complementary Items

Unbounded Variables in GAMS	Equation Name in GAMS	Equation Number in Model Framework
Xs(i,r)	EQ_MMs(i)	(22)
Xm(i,r)	EQ_MMm(i)	(23)
Xl(i,r)	EQ_MMo(i)	(24)
Xb(r)	EQ_MMb	(25)

The definition of subscripts, parameters, and variables in the model, as well as their corresponding GAMS item are depicted in Table 1. The variables in Table 3 are declared in GAMS using the statement below:

```
Positive Variables
Ys(i,r)
          'Domestic production seeds'
          'Domestic production meals'
Ym(i,r)
Yl(i,r) 'Domestic production oils'
MRG(i,r) 'Crush margin'
CR(i,r) 'Crush seed demand'
         'Domestic production biofuels'
Yb(r)
MRGb(r) 'Biofuel margin'
CPb(r) 'Intermediate demand of composite oils by biofuel sector'
Cm(i,f,r) 'Final demand of meals'
CPh(r) 'Final demand of composite oils by households'
Clb(i,r) 'Intermediate demand of oils by biofuels sector by oil type'
Clh(i,r) 'Final demand of oils by households by type of oil'
PCb(r) 'Price of composite oil consumed by biofuel sector'
PCh(r) 'Price of composite oil consumed by households'
Cb(f,r) 'Final demand of biofuels'
Ps(i,r) 'Domestic price seeds'
Pm(i,r) 'Domestic price meals'
Pl(i,r) 'Domestic price oils'
         'Domestic price biofuels'
Pb(r)
PCIFs(i) 'International price of seeds'
PCIFm(i) 'International price of meals'
PCIFl(i) 'International price of oils'
PCIFb 'International price of biofuels'
Variables
Xs(i,r) 'Net trade of seeds'
Xm(i,r) 'Net trade of meals'
Xl(i,r) 'Net trade of oils'
           'Net trade of biofuels'
Xb(r)
;
```

The inequalities and equations are declared in GAMS as follows:

Equations

EQ_Ys	seed	production
EQ_Ym	meal	production

EQ_Yo	oil production
EQ_MG	crush margin equation
EQ_CR	crush demand
EQ_Yb	biofuels production
EQ MGb	bfuel margin equation
EQ CPb	demand for composite oils (biofuel)
	final demand meals
EQ_CPh	demand for composite oils (households)
EQ_Cob	final demand oils (biofuel)
EQ_Coh	final demand oils (households)
EQ_PCb	price of composite oil (biofuel)
EQ_PCh	price of composite oil (households)
EQ_Cb	final demand biofuels
EQ_Ps	market clearing condition seeds
	market clearing condition meals
EQ_Po	market clearing condition oils
	market clearing condition biofuels
EQ_MMs	market clearing condition international trade seeds
EQ_MMm	market clearing condition international trade meals
EQ_MMo	market clearing condition international trade oils
EQ_MMb	market clearing condition international trade biofuels
	import price equation seeds
EQ_Xm	import price equation meals
EQ_Xo	import price equation oils
EQ_Xb	import price equation biofuels
;	

The inequalities and equations are specified in GAMS as follows. The function of each inequality or equation follows its name by two periods and spaces.

```
**Supply
EQ_Ys(i,r).. Ys(i,r) =g= Ks(i,r)*Area(i,r)*(Ps(i,r)**SIGA(i,r));
EQ_Ym(i,r).. Ym(i,r) =g= CR(i,r)*alpha("meal",i,r);
EQ_Yo(i,r).. Yl(i,r) =g= CR(i,r)*alpha("oil",i,r);
EQ_MG(i,r).. MRG(i,r) =g= (Pl(i,r)*alpha("oil",i,r)+
Pm(i,r)*alpha("meal",i,r))*(1+SUBs(i,r))-Ps(i,r);
EQ_CR(i,r).. CR(i,r) =g= A(i,r)*MRG(i,r)**CRMG(i,r);
EQ_Yb(r).. CPb(r) =g= Yb(r);
EQ_MG(i,r).. MRG(i,r) =g= (Pl(i,r)*alpha("oil",i,r)+
Pm(i,r)*alpha("meal",i,r))*(1+SUBs(i,r))-Ps(i,r);
EQ_CPb(r).. CPb(r) =g= Ko("Bio",r)*MRGb(r)**CRMG("soy",r);
***Demand
EQ_Cm(i,f,r).. Cm(i,f,r) =g= Km(i,f,r)*Pm(i,r)**SIGM(i,r);
EQ_CPh(r).. CPh(r) =g= Ko("H",r)*PCh(r)**SIGL(r);
```

```
EQ Cob(i, r).. Clb(i, r) =g=
a \overline{C}(i, "bio", r) * CPb(r) * (PCb(r) / Pl(i, r)) * * SIGO(i, "bio", r);
EQ Coh(i,r).. Clh(i,r) =g= a C(i,"h",r)*CPh(r)*(PCh(r)/Pl(i,r))**SIGO(i,"h",r);
EQ PCb(r). PCb(r) *CPb(r) = g = SUM(i, Clb(i, r) * Pl(i, r));
EQ PCh(r). PCh(r) * CPh(r) = q = SUM(i, Clh(i, r) * Pl(i, r));
EQ Cb(f,r).. Cb(f,r) =g= Kb(f,r)*Pb(r)**SIGB(r);
**Trade
EQ Ps(i,r).. Ys(i,r) = g = CR(i,r) + Xs(i,r);
EQ Pm(i,r).. Ym(i,r) = q = SUM(f,Cm(i,f,r)) + Xm(i,r);
EQ Po(i,r).. Yl(i,r) = g = Clb(i,r) + Clh(i,r) + Xl(i,r);
EQ Pb(r)..
             Yb(r) = g = SUM(f, Cb(f, r)) + Xb(r);
EQ Xs(i,r).. Ps(i,r) =q= PCIFs(i)*(1+timp("seed",i,r))/(1+texp("seed",i,r));
EQ Xm(i,r).. Pm(i,r) =g= PCIFm(i)*(1+timp("meal",i,r))/(1+texp("meal",i,r));
EQ Xo(i,r).. Pl(i,r) =g= PCIFl(i)*(1+timp("oil",i,r))/(1+texp("oil",i,r));
EQ Xb(r)..
             Pb(r) =g=
PCIFb*(1+timp("bfuel", "soy", r))/(1+texp("bfuel", "soy", r));
EQ MMs(i).. SUM(r,Xs(i,r)) === 0;
EQ MMm(i)..
             SUM(r, Xm(i, r)) = e = 0;
EQ MMo(i)..
             SUM(r, Xl(i, r)) = e = 0;
EQ MMb.. SUM(r, Xb(r)) = e = 0;
```

3.2.3 Declare the options for the model resolution

The *.scaleopt* command sets the scale option to 1 so as to turn on the scaling feature in GAMS. It is basically used to scale the percentage change of commodity domestic production in the simulation outputs. This statement should be inserted somewhere after the model statement and before the solve statement (Rosenthal 2011). The options of *limrow=100* and *limcol=100* limit the number of rows and columns listed in the output file (.lst format) to 100, so as to reduce the size of output file. The *MCP=PATH* command tells GAMS to use PATH solver to solve this MCP problem. The *"solve OSDS using MCP"* command tells GAMS to solve the OSDS model as a mixed complementary program.

OSDS.scaleopt = 1;

```
option limrow=100;
option limcol=100;
option MCP=PATH;
Solve OSDS using MCP;
```

3.2.4 Simulations of trade policy in the form of export taxes and import tariffs

The implementations of trade-distorting policies, such as export taxes and import tariffs, have different effects on the agricultural productions in different countries and commodities' international prices. The removal of export taxes of a commodity is expected to raise its domestic price and decrease its international price, which will encourage the domestic production. In addition, due to the different rates of export taxes that are applied along the oilseed value chain for different sectors, the removal of all export taxes have different effects on the productions of seeds, oils, and meals, and it might foster changes in the production structure. In contrast, the removal of import tariffs has the opposite effect on the commodity prices and production. For the counties in which both export taxes and import tariffs are implemented, the net effect of eliminating these two trade policies has an ambiguous effect on the price, production, and international trade flow. Thus, it is important to simulate the removal of export taxes and import tariffs on the oilseed products.

Based on the base case in which the influences of both export taxes and import tariffs are considered, six simulation scenarios are included in this study to analyze the effects of trade policy changes on the production, consumption, net export, price, and economic surplus for different commodity markets (Table 4). Export taxes on raw materials (e.g. seed products) have an effect equivalent to a production subsidy for downstream industries. Simulation scenario 4 and 5 are conducted to calculate the percentages of equivalent subsidies on seeds and oils.

Scenario	Description
Base Case	Export taxes and import tariffs are considered

Simulation Scenario 1	Remove all export taxes
Simulation Scenario 2	Remove import tariffs of major import countries (EU, US, and China)
Simulation Scenario 3	Remove all export taxes and import tariffs of major import countries (EU, US, and China)
Simulation Scenario 4	Subsidy equivalent to export taxes on seeds
Simulation Scenario 5	Subsidy equivalent to export taxes on oils
Simulation Scenario 6	Remove all export taxes and import tariffs

Simulation statements in GAMS are shown below. Parameter *TEXPsim(type, i,r,sim)* represents the export taxes of commodity *type* based on seed *i* that is imposed in region *r* for simulation scenario *sim*. The removal of export taxes is accomplished by assign a value of zero to this parameter, for example "*TEXPsim(type,i,r,'sim1') = 0*". Similarly, *TIMPsim(type, i,r,sim)* represents the import tariffs.

```
***Simulation scenenarios
Set
                    base case
removal all export taxes
sim /BASE
        SIM1
                    removal all export taxes
removal import tariffs (EU US and China)
removal export taxes and import tariffs (EU US and China)
subsidy equivalent to export taxes on seeds
subsidy equivalent to export taxes on oils
        SIM2
        SIM3
        SIM4
        SIM5
        SIM6
                     removal of all export taxes and tariffs
        1
simcur(sim) current simulation
SIMCUR(SIM)
                          = NO;
SIMCUR(SIM)
                           = YES;
Parameter
TEXPsim(type,i,r,sim) export taxes
TIMPsim(type,i,r,sim) import tariffs
MRGsim(i,r,sim)crush marginMRGbsim(i,r,sim)biodiesel marginYmsim(i,r,sim)domestic meal productionYlsim(i,r,sim)domestic oil productionYbsim(r,sim)domestic biodiesel production
TEXPsim(type,i,r,sim) = TEXP(type,i,r);
TIMPsim(type,i,r,sim) = TIMP(type,i,r);
MRGsim(i,r,sim) = MRGO(i,r);
MRGbsim(r,sim) = MRGbO(r);
MRGsim(1,1,5+...,
MRGbsim(r,sim) = MRGbO(r);
Ymsim(i,r,sim) = YO("meal",i,r);
Vlsim(i,r,sim) = YO("oil",i,r);
- VO("bfuel","soy
                              = YO("bfuel", "soy", r);
 *SIM1: removal all export taxes
TEXPsim(type,i,r,'sim1') = 0;
 *SIM2: removal import tariffs of EU US and China
TIMPsim(type,i,'EU','sim2') = 0;
```

```
TIMPsim(type,i,'USA','sim2') = 0;
TIMPsim(type,i,'CHN','sim2') = 0;
*SIM3: removal all export taxes and import tariffs of EU US and China
TEXPsim(type,i,r,'sim3') = 0;
TIMPsim(type,i,'EU','sim3') = 0;
TIMPsim(type,i,'USA','sim3') = 0;
TIMPsim(type,i,'CHN','sim3') = 0;
*SIM4: compute the subsidy equivalent to export taxes on seeds
TEXPsim("seed",i,r,'sim4') = 0;
MRGsim (i,r,'sim4') = MRGsim(i,r,'base');
Ymsim(i,r,sim) = Ymsim(i,r,"base");
Ylsim(i,r,sim) = Ylsim(i,r,"base");
*SIM5: compute the subsidy equivalent to export taxes on oil
TEXPsim("oil", i, r, 'sim5') = 0;
MRGbsim(r, 'sim5') = MRGbsim(r, 'base');
Ybsim(r,sim) = Ybsim(r,"base");
*SIM6: removal of all export taxes and import tariffs
TEXPsim(type,i,r,'sim6') = 0;
TIMPsim(type,i,'EU','sim6') = 0;
TIMPsim(type,i,'USA','sim6') = 0;
TIMPsim(type,i,'CHN','sim6') = 0;
TIMPsim(type,i,'ARG','sim6') = 0;
TIMPsim(type,i,'MYA','sim6') = 0;
TIMPsim(type,i,'IDN','sim6') = 0;
;
```

3.2.5 Interpret GAMS output

The model solution of the base case is displayed in the .lst output file, which contains the optimal values of all decision variables as follows:

model_sep_bf_mcp(Monica's Note).gms			arinit_mcp_	bf.gms	dat	data.gdx		model_sep_bf_mcp(Monica's Note).lst		
res_mcp.gdx			results_mcp_bf_Monica.gms			sim_mcp_bf_Monica.gms				
- Compilation		•	IND				+INF			
 Include File Summ 			IDN				+INF	42	.013	
 Equation Listing Equation 	SOLVE OSDS Using	g MC	EU				+INF	-1296	.540	
	SOLVE OSDS Using	мс	MYA				+INF	33	.007	
- Column	-	,	UKR				+INF			
	SOLVE OSDS Using I		CAN				+INF	42	.008	
- Solution Report : - SolEQU	SOLVE OSDS Using	MCI	ROW				+INF	-60	.987	
- SolVAR										
- Ys				VAR Ys	Domes	stic produ	ction	seeds		
- Ym - Yl										
MRG		E			LOWER	LEVEI	. т	JPPER	MARGINAL	
- CR										
- Yb			sov	.ARG		45059.06	57	+INF		
- MRGb - CPb			-	BRA		51559.07	12	+INF		
- Cm			-	.USA		75864.10		+INF		
- CPh			-	.CHN		12634.01	8	+INF		
- Clb				.IND		9968.01		+INF		
– Clh – PCb			-	.IDN		592.00		+INF		
- PCh			soy			1211.00		+INF		
- Cb			-	.MYA			-	+INF		
– Ps – Pm			-	.UKR		813.00	1	+INF		
- PI			-	.CAN		3396.00		+INF		
- Pb			-	.ROW		11322.02		+INF		
- PCIFs			-	ARG		3594.00		+INF		
 PCIFm PCIFI 				.BRA		105.00		+INF		
- PCIFb				USA	•	1253.00	-	+INF	•	
- Xs				.CHN	•	1187.00		+INF	•	
- Xm				. IND	•	1463.00		+INF	•	
				. TDN	•	1403.00		+INF	•	

The model solutions of simulation scenarios are generated based on the above output. As mentioned previously, the file "*results_mcp.gms*" is used to calculate the percentage changes of variables in each simulation scenario based on the base case output. The associated parameters are divided into six categories: production, price, consumption, net trade, crush margin and economic surplus, as defined below:

Parameters * Production					
Ysxp(i,r,sim)	Percentage	change	of	domestic	seeds production volume
Ylxp(i,r,sim)	Percentage	change	of	domestic	oils production volume
Ymxp(i,r,sim)	Percentage	change	of	domestic	meals production volume
Ybxp(r,sim)	Percentage	change	of	domestic	biodiesel production volume
Ysvalxp(i,r,sim)	Percentage	change	of	domestic	<pre>seeds production value(\$)</pre>
Ylvalxp(i,r,sim)	Percentage	change	of	domestic	oils production value(\$)
Ymvalxp(i,r,sim)	Percentage	change	of	domestic	<pre>meals production value(\$)</pre>
Ybvalxp(r,sim) * Price	Percentage	change	of	domestic	<pre>biodiesel production value(\$)</pre>
Psxp(i,r,sim) Plxp(i,r,sim)					seeds price oils price
TTVP (T, T, STW)	rereentage	change	OT.	uomestre	OTTO PITCO

Percentage change of domestic meals price Pmxp(i,r,sim) Percentage change of domestic biodiesel price Pbxp(r,sim) PChxp(r,sim) Percentage change of domestic price of composite oils by household PCbxp(r,sim) Percentage change of domestic price of composite oils by biodiesel PCIFsxp(i,sim) Percentage change of international price of seeds PCIFlxp(i,sim) Percentage change of international price of oils PCIFmxp(i,sim) Percentage change of international price of meails PCIFbxp(sim) Percentage change of international price of biodiesel * Consumption Crxp(i,r,sim) Percentage change of intermediate consumption volume of seeds Clbxp(i,r,sim) Percentage change of final consumption volume of oils by household Percentage change of final consumption volume of oils by biodiesel Clhxp(i,r,sim) Percentage change of final consumption volume of composite oils by CPhxp(r,sim) household sector CPbxp(r,sim) Percentage change of final consumption volume of composite oils by biodiesel sector Cmxp(i,f,r,sim) Percentage change of final consumption volume of meals Cbxp(f,r,sim) Percentage change of final consumption volume of biodiesel Percentage change of intermediate consumption value (\$) of seeds CRvalxp(i,r,sim) Clhvalxp(i,r,sim) Percentage change of final consumption value(\$) of oils by household sector Percentage change of final consumption value(\$) of oils by Clbvalxp(i,r,sim) biodiesel sector CPhvalxp(r,sim) Percentage change of final consumption value(\$) of composite oils by household sector Percentage change of final consumption value(\$) of composite oils CPbvalxp(r,sim) by biodiesel sector Percentage change of final consumption value(\$) of meals Cmvalxp(i,f,r,sim) Cbvalxp(f,r,sim) Percentage change of final consumption value(\$) of biodiesel * Net trade Xsxp(i,r,sim) Percentage change of net trade volume of seeds from region r Xlxp(i,r,sim) Percentage change of net trade volume of oils from region r Xmxp(i,r,sim) Percentage change of net trade volume of seeds from region r Percentage change of net trade volume of biodiesel from region r Xbxp(r,sim) Xsvalxp dp(i,r,sim) Percentage change of net trade value(\$) of seeds from region r base on domestic price Xlvalxp dp(i,r,sim) Percentage change of net trade value(\$) of oils from region r base on domestic price Xmvalxp dp(i,r,sim) Percentage change of net trade value(\$) of meals from region r base on domestic price Percentage change of net trade value(\$) of biodiesel from region r Xbvalxp dp(r,sim) base on domestic price Xsvalxp_ip(i,r,sim) Percentage change of net trade value(\$) of seeds from region r base on international price Xlvalxp ip(i,r,sim) Percentage change of net trade value(\$) of oils from region r base on international price Xmvalxp ip(i,r,sim) Percentage change of net trade value(\$) of meals from region r base on international price Percentage change of net trade value(\$) of biodiesel from region r Xbvalxp ip(r,sim) base on international price * Crush Margin MRGxp(i,r,sim) Percentage change of crush margin Percentage change of biodiesel crush margin MRGbxp(r,sim) * Economic Surplus Csurplus(r,sim) Variation in consumer surpluses from removal of export taxes on oils Psurplus(r,sim) Variation in producer surpluses from removal of export taxes on oils Harberger triangle (deadweight loss associated with trade Harberger(r,sim) intervention) ;

```
22
```

Parameters in the former five categories are directly generated from the percentage value changes of decision variables, while parameters in the economic surplus category are calculated based on the formulas of consumer surplus, producer surplus, and deadweight loss (Harberger's triangle).

The following statement is an example of calculating the percentage change of domestic production volume for seeds, oils, meals, and biodiesel, respectively.

```
Ysxp(i,r,sim)$Ysres(i,r,'BASE') = (Ysres(i,r,sim)/Ysres(i,r,'BASE')-1)*100;
Ylxp(i,r,sim)$Ylres(i,r,'BASE') = (Ylres(i,r,sim)/Ylres(i,r,'BASE')-1)*100;
Ymxp(i,r,sim)$Ymres(i,r,'BASE') = (Ymres(i,r,sim)/Ymres(i,r,'BASE')-1)*100;
Ybxp(r,sim)$Ybres(r,'BASE') = (Ybres(r,sim)/Ybres(r,'BASE')-1)*100;
```

The following statement is an example of calculating the value change in consumer surplus *Csurplus(r,sim)* and producer surplus *Psurplus(r,sim)* from removal of export taxes on oils for each country, as well as the deadweight loss *Harberger(r,sim)* associated with implementing the tax policies in domestic market.

```
Csurplus(r,sim) = -((CPhres(r,sim)*(PChres(r,sim) -
PChres(r,"BASE"))+((CPhres(r,"BASE")-CPhres(r,sim))*((PChres(r,sim) -
PChres(r,"BASE"))))+(CPbres(r,sim)*(PCbres(r,sim)-PCbres(r,"BASE"))+
((CPbres(r,"BASE")-CPbres(r,sim))*((PCbres(r,sim)-PCbres(r,"BASE")))))))/2 ;
Psurplus(r,sim) = sum(i, Ylres(i,r,sim)*(Plres(i,r,sim)-Plres(i,r,'base')) -
((Ylres(i,r,sim)-Ylres(i,r,'base'))*(Plres(i,r,sim)-Plres(i,r,'base')))/2);
Harberger(r,sim) = (((CPhres(r,"BASE")-CPhres(r,sim))*((PChres(r,sim)-
PChres(r,"BASE")))+(CPbres(r,"BASE")-CPhres(r,sim))*((PCbres(r,sim)-
PCbres(r,"BASE")))/2) + sum(i, (Ylres(i,r,sim)-
Ylres(i,r,'base'))*(Plres(i,r,sim)-Plres(i,r,'base'))/2);
```

Values of the above parameters are loaded into the .gdx file "*res_mcp.gdx*". For example, the following graph shows the value of parameter Ybxp(r, sim), the percentage change of domestic biodiesel production volume, in different simulation scenario. The yellow highlighted figure represents that US's domestic biodiesel production will increase by 6.59% if all the export taxes and import tariffs are eliminated in global extent.

model_sep_bf_mcp(Monica's Note).gms						varinit_mcp_bf.gms			data.gdx		model_sep_bf_mcp(Monica's		
res_mcp.gdx						results_mcp_bf_Monica.gms					sim_mcp_bf		
Entry	Symbol	Туре	Dim	Nr Elem	*	Ybxp							
1	Ysxp	Par	3	204									
2	Ylxp	Par	3	210									
3	Ymxp	Par	3	210									
4	Ybxp	Par	2	54								Pla	ane Ind
5	Psxp	Par	3	264									
6	Plxp	Par	3	264									
7	Pmxp	Par	3	264		L							
8	Pbxp	Par	2	66			SIM1	SIM2	SIM3	SIM4	SIM5	SIM6	
9	PChxp	Par	2	66		ARG	-0.36097	-0.95730	-1.65226	-0.18811	-0.01075	-1.74366	
10	PCbxp	Par	2	54		BRA	0.24804	-0.63634	-0.28578	-0.12052	-0.00609	-0.31885	
11	Сгхр	Par	3	210		USA	1.14055	5.76268	6.65789	-0.37643	0.04198	6.59241	
12	Clhxp	Par	3	252		CHN	0.89383	-0.95509	0.06554	-0.16713	-0.01307	0.01776	
13	Clbxp	Par	3	90		IDN	-0.91742	0.07803	-0.84368	0.03801	-0.00194	-1.51908	
14	CPhxp	Par	2	66		EU	0.02749	-0.13263	-0.09286	-0.00763	-0.00083	-0.09264	
15	CPbxp	Par	2	54		MYA	-1.40637	0.08489	-1.33173	0.03482	-0.00179	-2.18847	
16	Стхр	Par	4	222		CAN	0.39273	-0.95448	-0.37344	-0.13557	-1.90124	-0.36046	
17	CRvalxp	Par	3	210		ROW	0.40935	-0.23069	0.25173	-0.00535	0.82961	0.24033	
18	Clhvalxp	Par	3	252									
19	Clbvalxp	Par	3	90									
20	CPhyalxp	Par	2	66									

4 Appendix A, GAMS code of main file erato_mcp.gms

** Partial equilibrium model for oilseeds market. 11 countries ** Four types of commodities: oilseeds, meals and oils, and biofuels. ** Four oilseeds: soybean, rapeseeds, sunflower, palm nuts ** Revised on Dec 6, 2011 option limrow = 1; **Set definition Sets type product type /seed meal oil bfuel / i sectors /soy sunf rape palm / /ARG Argentina r countries BRA Brazil United States USA CHN China IND India IDN Indonesia European Union ΕU Malaysia MYA UKR Ukraine CAN Canada Rest of the world / ROW f final consumers /H Households Bio Bioenergy/ alias (r,s), (i,j); **Calibration of the model \$include .\variablesbase bf.gms \$call "GDXXRW .\data_oct_bf.xls O=.\data.gdx Index=Index!A1:F11" PARAMETER Yinit International Production ICinit Intermediate Consumption?? Cinit Xinit International Net Export elast prices dd exptax areas PARAMETER *---- TAXES AND SUBVENTIONS ----TEXP(type, i, r) 'Export tax rate' TIMP(type,i,r) 'Import duties'

```
*---- PARAMETERS ----
Area(i,r)
              'Area'
               'Supply area elasticity'
SIGA(i,r)
               'Crush margin elasticity'
CRMG(i,r)
               'Elasticity of substitution among oils'
SIGO(i,f,r)
SIGL(r)
               'Price elasticity of demand of oils'
SIGM(i,r)
               'Price elasticity of demand of meals'
SIGB(r)
               'Price elasticity of demand of biodiesel'
Ks(i,r)
               'Scale parameter in isoelastic supply functions'
Ko(f,r)
               'Scale parameter in isoelastic demand functions (oils)'
              'Scale parameter in isoelastic demand functions (meals)'
Km(i,f,r)
              'Scale parameter in isoelastic demand functions (biodiesel)'
Kb(f,r)
alpha(type,i,r)'Coefficient for production of oils and meals'
              'Scale parameter in composite oil demand'
a C(i,f,r)
               'Subsidy equivalent for production of oils and seeds'
SUBs(i,r)
SUBb(r)
               'Subsidy equivalent for production of biofuel'
A(i,r)
               'Scale parameter crush demand'
;
$GDXIN .\data.gdx
$LOAD Yinit
$LOAD ICinit
$LOAD Cinit
$LOAD Xinit
$LOAD elast
$LOAD prices
$LOAD dd
$LOAD exptax
$LOAD areas
$GDXIN
**Calibration
Area(i,r) = Areas(i,r,"Ha");
TEXP(type,i,r) = exptax(type,i,r);
TIMP(type,i,r) = dd(type,i,r);
SIGA(i,r)
                = 0.3;
SIGA(i,r)
                = elast(i,r,'Area');
SIGL(r)
                = -0.38;
                = -0.28;
SIGL('ARG')
SIGO(i,"h",r) = 2;
SIGO(i,"bio",r) = 2;
               = -0.3;
SIGM(i,r)
               = elast(i,r,'Mealp');
SIGM(i,r)
SIGB(r)
               = -0.25;
                = 0;
SUBs(i,r)
SUBb(r)
               = 0;
                    = Yinit(type,i,r);
YO(type,i,r)
                    = ICinit(i,r,"seed");
CRO(i,r)
CO(type, i, f, r)
                    = Cinit(type, i, f, r);
CPO(f,r)
                    = sum(i,CO("oil",i,f,r));
XO(type,i,r)
                    = Xinit(type,i,r);
CRMG(i,r)
                                  = elast(i,r,'Crush');
alpha(type,i,r)
                                  = 1;
alpha("oil",i,r) $CRO(i,r) = YO("oil",i,r)/CRO(i,r);
alpha("meal",i,r)$CRO(i,r) = YO("meal",i,r)/CRO(i,r);
PCIFO(type,i)
                   = prices(type,i);
```

```
= PCIFO("oil",i)*0.35;
PCIFO("seed",i)
PO(type,i,r)
                    = PCIFO(type,i)/((1+TEXP(type,i,r))/(1+TIMP(type,i,r)));
PCO(f,r) $CPO(f,r)
                             = sum(i,CO("oil",i,f,r)*PO("oil",i,r))/CPO(f,r);
                     = YO("seed",i,r)/(Area(i,r)*(PO("seed",i,r)**SIGA(i,r)));
Ks(i,r)
Ko("h",r)$PCO("h",r) = CPO("h",r)/(PCO("h",r)**SIGL(r));
Km(i,f,r)
                     = CO("meal",i,f,r)/(PO("meal",i,r)**SIGM(i,r));
a C(i, "bio", r) $CPO("bio", r) =
CO("oil",i,"bio",r)/(CPO("bio",r)*(PCO("bio",r)/PO("oil",i,r))**SIGO(i,"bio",r));
a C(i,"h",r)$CPO("h",r)
CO("oil",i,"h",r)/(CPO("h",r)*(PCO("h",r)/PO("oil",i,r))**SIGO(i,"h",r));
Kb(f,r)$PO("bfuel","soy",r) = CO("bfuel","soy",f,r)/(PO("bfuel","soy",r)**SIGB(r));
MRGO(i,r)
(PO("meal",i,r)*alpha("meal",i,r)+PO("oil",i,r)*alpha("oil",i,r))*(1-SUBs(i,r))-
PO("seed",i,r);
A(i,r) $MRGO(i,r)
                   = CRO(i,r) / (MRGO(i,r) * CRMG(i,r));
                    = PO("bfuel", "soy", r) * (1-SUBb(r)) - PCO("Bio", r);
MRGbO(r)
Ko("Bio",r)$MRGbO(r) = CPO("Bio",r)/MRGbO(r)**CRMG("soy",r);
;
Positive Variables
        'Domestic production seeds'
Ys(i,r)
        'Domestic production meals'
Ym(i,r)
Yl(i,r)
        'Domestic production oils'
MRG(i,r) 'Crush margin'
CR(i,r) 'Crush seed demand'
         'Domestic production biofuels'
Yb(r)
MRGb(r) 'Biofuel margin'
CPb(r)
         'Intermediate demand of composite oils by biofuel sector'
Cm(i,f,r) 'Final demand of meals'
CPh(r)
         'Final demand of composite oils by households'
         'Intermediate demand of oils by biofuels sector by oil type'
Clb(i,r)
Clh(i,r) 'Final demand of oils by households by type of oil'
          'Price of composite oil consumed by biofuel sector'
PCb(r)
        'Price of composite oil consumed by households'
PCh(r)
Cb(f,r) 'Final demand of biofuels'
        'Domestic price seeds'
Ps(i,r)
        'Domestic price meals'
Pm(i,r)
Pl(i,r) 'Domestic price oils'
         'Domestic price biofuels'
Pb(r)
PCIFs(i) 'International price of seeds'
PCIFm(i) 'International price of meals'
PCIF1(i) 'International price of oils'
         'International price of biofuels'
PCIFb
Variables
Xs(i,r) 'Net trade of seeds'
          'Net trade of meals'
Xm(i,r)
Xl(i,r)
          'Net trade of oils'
          'Net trade of biofuels'
Xb(r)
;
Equations
EO Ys
           seed production
EO Ym
          meal production
EQ Yo
           oil production
Е
                         equation
```

EQ_MG	crush	margin	е
EQ_CR	crush	demand	

EQ Yb biofuels production EQ MGb bfuel margin equation EQ CPb demand for composite oils (biofuel) EQ Cm final demand meals EQ CPh demand for composite oils (households) EQ Cob final demand oils (biofuel) EQ Coh final demand oils (households) EQ_PCb price of composite oil (biofuel) EQ_PCh price of composite oil (households) EQ Cb final demand biofuels EQ Ps market clearing condition seeds EQ Pm market clearing condition meals EQ Po market clearing condition oils EQ Pb market clearing condition biofuels EQ MMs market clearing condition international trade seeds market clearing condition international trade meals EQ MMm EQ MMo market clearing condition international trade oils EQ_MMb market clearing condition international trade biofuels EQ Xs import price equation seeds EQ_Xm import price equation meals EQ_Xo import price equation oils EQ_Xb import price equation biofuels ; **Supply Ys(i,r) =g= Ks(i,r)*Area(i,r)*(Ps(i,r)**SIGA(i,r)); EQ Ys(i,r).. EQ Ym(i,r).. Ym(i,r) =g= CR(i,r)*alpha("meal",i,r); EQ Yo(i,r).. Yl(i,r) =g= CR(i,r) *alpha("oil",i,r); EQ MG(i,r).. MRG(i,r) =g= (Pl(i,r)*alpha("oil",i,r)+ Pm(i,r)*alpha("meal",i,r))*(1+SUBs(i,r))-Ps(i,r); EQ CR(i,r).. CR(i,r) =g= A(i,r)*MRG(i,r)**CRMG(i,r); EQ Yb(r).. CPb(r) = q = Yb(r);EQ MGb(r).. MRGb(r) = g = Pb(r) * (1 + SUBb(r)) - PCb(r);EQ CPb(r).. CPb(r) =g= Ko("Bio", r) *MRGb(r) **CRMG("soy", r); **Demand EQ Cm(i,f,r).. Cm(i,f,r) =g= Km(i,f,r)*Pm(i,r)**SIGM(i,r); EQ CPh(r).. CPh(r) =g= Ko("H",r)*PCh(r)**SIGL(r); EQ Cob(i,r).. Clb(i,r) =g= a C(i,"bio",r)*CPb(r)*(PCb(r)/Pl(i,r))**SIGO(i,"bio",r); EQ Coh(i,r).. Clh(i,r) =g= a C(i,"h",r)*CPh(r)*(PCh(r)/Pl(i,r))**SIGO(i,"h",r); EQ PCb(r). PCb(r) *CPb(r) = q = SUM(i, Clb(i, r) * Pl(i, r));EQ PCh(r).. PCh(r) *CPh(r) =g= SUM(i, Clh(i, r) *Pl(i, r));EQ Cb(f,r).. Cb(f,r) = g= Kb(f,r) * Pb(r) * * SIGB(r); **Trade EQ Ps(i,r).. Ys(i,r) = g = CR(i,r) + Xs(i,r);

EQ_Pm(i,r).. Ym(i,r) =g= SUM(f,Cm(i,f,r)) + Xm(i,r); EQ Po(i,r).. Yl(i,r) = g = Clb(i,r) + Clh(i,r) + Xl(i,r);Yb(r) = g = SUM(f, Cb(f, r)) + Xb(r);EQ Pb(r).. Ps(i,r) =g= PCIFs(i)*(1+timp("seed",i,r))/(1+texp("seed",i,r)); EQ Xs(i,r).. EQ_Xm(i,r).. Pm(i,r) =g= PCIFm(i)*(1+timp("meal",i,r))/(1+texp("meal",i,r)); EQ Xo(i,r).. Pl(i,r) =g= PCIFl(i)*(1+timp("oil",i,r))/(1+texp("oil",i,r)); Pb(r) =g= PCIFb*(1+timp("bfuel","soy",r))/(1+texp("bfuel","soy",r)); EQ Xb(r).. EQ MMs(i).. SUM(r, Xs(i, r)) = e = 0;EQ MMm(i).. SUM(r,Xm(i,r)) =e= 0; EQ MMo(i).. SUM(r, Xl(i, r)) = e = 0;EQ_MMb.. SUM(r, Xb(r)) = e = 0;

Model OSDS / EQ Ys.Ys EQ Ym.Ym EQ Yo.Yl EQ MG.MRG EQ CR.CR EQ_Yb.Yb EQ_MGb.MRGb EQ_CPb.CPb $EQ_Cm.Cm$ EQ_CPh.CPh EQ Cob.Clb EQ Coh.Clh EQ PCb.PCb EQ_PCh.PCh EQ_Cb.Cb EQ Ps.Ps EQ Pm.Pm EQ Po.Pl EQ_Pb.Pb EQ_MMs.PCIFs EQ MMm.PCIFm EQ_MMo.PCIF1 EQ MMb.PCIFb EQ Xs.Xs

\$include .\varinit mcp bf.gms

EQ_Xb.Xb /;

EQ_Xm.Xm EQ_Xo.Xl

osds.scaleopt = 1;

option limrow=100;

option limcol=100; option MCP=PATH ; solve osds using mcp; \$include sim_mcp_bf_Monica.gms

\$include results_mcp_bf.gms

5 Appendix B, GAMS code of simulation file sim_mcp.gms

***Simulation scenenarios

```
Set
sim /BASE
                    base case
                  removal all export taxes
      SIM1
                 removal import tariffs (EU US and China)
removal export taxes and import tariffs (EU US and China)
subsidy equivalent to export taxes on seeds
subsidy equivalent to export taxes on oils
removal of all export taxes and tariffs
      SIM2
      STM3
      SIM4
      STM5
      SIM6
      /
simcur(sim) current simulation
SIMCUR(SIM)
                  = NO;
SIMCUR(SIM)
                     = YES;
Parameter
FarameterTEXPsim(type,i,r,sim)export taxesTIMPsim(type,i,r,sim)import tariffsMRGsim(i,r,sim)crush marginMRGbsim(r,sim)biodiesel marginYmsim(i,r,sim)domestic meal productionYlsim(i,r,sim)domestic oil productionYbsim(r,sim)domestic biodiesel production
TEXPsim(type,i,r,sim) = TEXP(type,i,r);
TIMPsim(type,i,r,sim) = TIMP(type,i,r);
MRGsim(i,r,sim) = MRGO(i,r);
                        = MRGbO(r);
MRGbsim(r,sim)
Ymsim(i,r,sim)
                      = YO("meal",i,r);
= YO("oil",i,r);
Ylsim(i,r,sim)
Ybsim(r,sim)
                          = YO("bfuel", "soy", r);
*SIM1: removal all export taxes
TEXPsim(type,i,r,'sim1') = 0;
*SIM2: removal import tariffs of EU US and China
TIMPsim(type,i,'EU','sim2') = 0;
TIMPsim(type,i,'USA','sim2') = 0;
TIMPsim(type,i,'CHN','sim2') = 0;
*SIM3: removal all export taxes and import tariffs of EU US and China
TEXPsim(type,i,r,'sim3') = 0;
TIMPsim(type,i,'EU','sim3') = 0;
TIMPsim(type,i,'USA','sim3') = 0;
TIMPsim(type,i,'CHN','sim3') = 0;
*SIM4: compute the subsidy equivalent to export taxes on seeds
TEXPsim("seed",i,r,'sim4') = 0;
MRGsim (i,r,'sim4') = MRGsim(i,r,'base');
Ymsim(i,r,sim) = Ymsim(i,r,"base");
Ylsim(i,r,sim) = Ylsim(i,r,"base");
*SIM5: compute the subsidy equivalent to export taxes on oil
TEXPsim("oil",i,r,'sim5') = 0;
MRGbsim(r,'sim5') = MRGbsim(r,'base');
Ybsim(r,sim) = Ybsim(r,"base");
*SIM6: removal of all export taxes and import tariffs
```

```
TEXPsim(type,i,r,'sim6')
                           = 0:
TIMPsim(type,i,'EU','sim6') = 0;
TIMPsim(type,i,'USA','sim6') = 0;
TIMPsim(type,i,'CHN','sim6') = 0;
TIMPsim(type,i,'ARG','sim6') = 0;
TIMPsim(type,i,'MYA','sim6') = 0;
TIMPsim(type,i,'IDN','sim6') = 0;
;
PARAMETER
                  'Domestic production of seeds'
Ysres(i,r,sim)
Ylres(i,r,sim)
                  'Domestic production of oils'
Ymres(i,r,sim)
                  'Domestic production of meals'
                  'Domestic production of biodiesel'
Ybres(r,sim)
Psres(i,r,sim)
                  'Domestic price of seeds'
Plres(i,r,sim)
                  'Domestic price of oils'
                  'Domestic price of meals'
Pmres(i,r,sim)
                  'Domestic price of biodiesel'
Pbres(r,sim)
PCbres(r,sim)
                  'Price of composite oil consumed by agent a'
PChres(r,sim)
                  'Price of composite oil consumed by agent a'
CRres(i,r,sim)
                  'Intermediate consumption of seeds'
Clbres(i,r,sim)
                  'Final demand of oils'
                 'Final demand of oils'
Clhres(i,r,sim)
                 'Final demand of composite oils'
CPbres(r,sim)
                 'Final demand of composite oils'
CPhres(r,sim)
Cmres(i,f,r,sim) 'Final demand of meals'
Cbres(f,r,sim)
                 'Final demand of biodiesel'
Xsres(i,r,sim)
                 'Net trade of seeds from r'
Xlres(i,r,sim)
                  'Net trade of oils from r'
Xmres(i,r,sim)
                  'Net trade of meals from r'
                  'Net trade of biodiesel from r'
Xbres(r,sim)
                  'International price of seeds'
PCIFsres(i,sim)
                  'International price of oils'
PCIFlres(i,sim)
                  'International price of meals'
PCIFmres(i,sim)
PCIFbres(sim)
                  'International price of biodiesel'
SUBsres(i,r,sim)
SUBbres(r,sim)
MRGres(i,r,sim)
                   'Crush margin'
MRGbres(r,sim)
LOOP(sim$simcur(sim),
$include varinit mcp bf.gms
TEXP(type,i,r) = TEXPsim(type,i,r,sim) ;
TIMP(type,i,r) = TIMPsim(type,i,r,sim) ;
*solve osds using cns;
solve osds using mcp;
Ysres(i,r,sim) = Ys.l(i,r)
                              ;
Ylres(i,r,sim) = Yl.l(i,r)
                              ;
Ymres(i,r,sim) = Ym.l(i,r)
                              ;
               = Yb.l(r)
Ybres(r,sim)
                               ;
Psres(i,r,sim) = Ps.l(i,r)
                              ;
Plres(i, r, sim) = Pl.l(i, r)
                              ;
Pmres(i,r,sim) = Pm.l(i,r)
                              ;
Pbres(r,sim)
              = Pb.l(r)
                              ;
PCbres(r, sim) = PCb.l(r)
                              ;
```

;

PChres(r, sim) = PCh.l(r)

```
CRres(i, r, sim) = CR.l(i, r)
                              ;
Clbres(i,r,sim) = Clb.l(i,r);
Clhres(i,r,sim) = Clh.l(i,r);
CPbres(r,sim) = CPb.l(r)
                               ;
CPhres(r, sim) = CPh.l(r)
                               ;
Cmres(i,f,r,sim) = Cm.l(i,f,r) ;
Cbres(f,r,sim) = Cb.l(f,r)
                              ;
Xsres(i,r,sim) = Xs.l(i,r)
Xlres(i,r,sim) = Xl.l(i,r)
                               ;
                               ;
Xmres(i,r,sim) = Xm.l(i,r)
                               ;
Xbres(r,sim)
               = Xb.l(r)
                               ;
PCIFsres(i,sim) = PCIFs.l(i)
                               ;
PCIFlres(i,sim) = PCIFl.l(i)
                               ;
PCIFmres(i,sim) = PCIFm.l(i)
                              ;
PCIFbres(sim) = PCIFb.1
                               ;
MRGres(i,r,sim) = MRG.l(i,r)
                               ;
MRGbres(r, sim) = MRGb.l(r)
                               ;
MRGres(i,r,'sim4') = MRGsim(i,r,'sim4')
                                          ;
MRGbres(r,'sim5') = MRGbsim(r,'sim5')
                                          ;
Ymres(i,r,'sim4') = Ymsim(i,r,"sim4");
Ylres(i,r,'sim4') = Ylsim(i,r,"sim4");
Ybres(r,'sim5') = Ybsim(r,"sim5");
SUBsres(i,r,'sim4')$(Plres(i,r,'sim4')*alpha("oil",i,r)+Pmres(i,r,'sim4')*alpha("meal"
,i,r)) =
(MRGres(i,r,'sim4')+Psres(i,r,'sim4'))/(Plres(i,r,'sim4')*alpha("oil",i,r)+Pmres(i,r,'
sim4')*alpha("meal",i,r))-1;
SUBbres(r,'sim5')$Pbres(r,'sim5')
                                  = (MRGbres(r,'sim5')
+PCbres(r, 'sim5'))/Pbres(r, 'sim5')-1;
);
```

6 Reference

FAO (*Food and Agriculture Organization*). 2007. "FAO Food Balance sheet". Available online: <u>http://faostat.fao.org/site/368/default.aspx#ancor</u>

FAPRI (Food and Agricultural Policy Research Institute). 2011. "FAPRI Elasticity Database". Available online: <u>http://www.fapri.iastate.edu/tools/elasticity.aspx</u>

Laborde, D. 2010 (miss the citation)

Lofgren, Hans, and Sherman Robinson. 1997. "The Mixed-complementarity Approach To specifying Agricultural Supply in Computable General Equilibrium Models."

McCarl, B.A. 2011. "McCarl Expanded GAMS User Guide Version 23.6" Available online: <u>http://gams.com/mccarl/mccarlhtml/index.html?mixed complementarity problem %28</u> <u>mcp%29.htm</u>

Rosenthal, R. E. 2011. "GAMS, A User's Guide." Washington, DC: GAMS Development Corporation. Available online: <u>http://www.gams.com/dd/docs/bigdocs/GAMSUsersGuide.pdf</u>

Rutherford, T. 1995. "Extension of GAMS for complementarity problems arising in applied economic analysis." *Journal of Economic Dynamics and Control* 19(8):1299-1324.